Invariance Problem

In order to visualize the invariance problem of the mean-centred first-order second-moment reliability method, considering the following limit-state functions:

```
g1 = R - S;
g2 = Log[R/S];
g3 = 1 - S/R;
```

Gradient vectors:

```
∇g1 = {D[g1, R], D[g1, S]};
∇g2 = {D[g2, R], D[g2, S]};
∇g3 = {D[g3, R], D[g3, S]};
```

Covariance matrix:

```
Σ = {{sR^2, ρRS sR sS}, {ρRS sR sS, sS^2}};
```

Second-moment information:

```
μR = 30;
μS = 20;
σR = 5;
σS = 10;
ρ = 0.5;
info = {R -> μR, S -> μS, sR -> σR, sS -> σS, ρRS -> ρ};
```

MCFOSM reliability indices:

```
β1 = (g1 / Sqrt[∇g1.Σ.∇g1]) /. info // N
```

which yields: 1.1547

```
β2 = ( g2 / Sqrt[∇g2.Σ.∇g2] ) /. info // N
```

which yields: 0.919508

```
β3 = ( g3 / Sqrt[∇g3.Σ.∇g3] ) /. info // N
```

which yields: 1.13389

Linearized limit-state functions:

```
g1Lin = (g1 + ∇g1.({r, s} - {R, S})) /. info /. {r -> R, s -> S};
g2Lin = g2 + ∇g2.({r, s} - {R, S}) /. info /. {r -> R, s -> S};
g3Lin = g3 + ∇g3.({r, s} - {R, S}) /. info /. {r -> R, s -> S};
```

Solve linearized limit-state functions for S:

```
s1 = Solve[g1Lin == 0, S];
s2 = Solve[g2Lin == 0, S];
s3 = Solve[g3Lin == 0, S];
```

Plot limit-state surfaces in the original variable space:

```
Plot[{S /. s1, S /. s2, S /. s3}, {R, 0, 1.5 μR},
 PlotStyle → {Green, Blue, Red},
 PlotLegends →
  Placed[
   LineLegend[{"g=R-S (linear)", "g=ln(R/S) (linearized nonlinear)",
     "g=1-S/R (linearized nonlinear)"}], {Left, Top}]]
```

which yields:



Prepare for transformation into standard space, by first establishing the mean vector:

```
M = {μR, μS};
```

Then the Cholesky decomposition of the covariance matrix:

```
L = CholeskyDecomposition[Σ /. info]ᵀ;
```

Second-moment transformation:

```
X = M + L.{Y1, Y2};
```

Transform limit-state functions into the standardized space:

```
G1 = g1 /. {R -> X[[1]], S -> X[[2]]};
G2 = g2 /. {R -> X[[1]], S -> X[[2]]}
G3 = g3 /. {R -> X[[1]], S -> X[[2]]};
```

which yields: $\text{Log}\left[\frac{30.+5.\,Y1}{20+5.\,Y1+8.66025\,Y2}\right]$

Gradient vectors:

```
∇G1 = {D[G1, Y1], D[G1, Y2]};
∇G2 = {D[G2, Y1], D[G2, Y2]}
∇G3 = {D[G3, Y1], D[G3, Y2]};
```

which yields: $\left\{\frac{(20+5.\,Y1+8.66025\,Y2)\left(-\frac{5.\,(30.+5.\,Y1)}{(20+5.\,Y1+8.66025\,Y2)^2}+\frac{5.}{20+5.\,Y1+8.66025\,Y2}\right)}{30.+5.\,Y1},\ -\frac{8.66025}{20+5.\,Y1+8.66025\,Y2}\right\}$

Linearize the limit-state functions in the standardized space:

```
G1Lin = (G1 + ∇G1.({y1, y2} - {Y1, Y2})) /. {Y1 -> 0, Y2 -> 0};
G2Lin = (G2 + ∇G2.({y1, y2} - {Y1, Y2})) /. {Y1 -> 0, Y2 -> 0};
G3Lin = (G3 + ∇G3.({y1, y2} - {Y1, Y2})) /. {Y1 -> 0, Y2 -> 0};
```

Solve standardized limit-state functions for Y2:

```
s1s = Solve[G1Lin == 0, y2];
s2s = Solve[G2Lin == 0, y2]
s3s = Solve[G3Lin == 0, y2];
```

which yields: $\{\{y2 \rightarrow -2.3094\,(-0.405465+0.0833333\,y1)\}\}$

Plot limit-state surfaces in the standardized variable space:

```
plotBound = 1.5;
Plot[{y2 /. s1s, y2 /. s2s, y2 /. s3s}, {y1, -plotBound, plotBound},
 PlotRange -> {{-plotBound, plotBound}, {-plotBound, plotBound}},
 PlotStyle → {Green, Blue, Red},
 PlotLegends → Placed[LineLegend[{"g=R-S", "g=ln(R/S)", "g=1-S/R"}],
   {Left, Bottom}]]
```

which yields: